Materialien zur
Einführung in Computeralgebrasysteme I (*Mathematica*)

Ralf Schaper

Wintersemester 2009 / 10

■ Einleitung

Mathematica wird von seinen Autoren und Herstellern bei Wolfram Research Inc. bezeichnet als "A System for Doing Mathematics by Computer". *Mathematica* enthält viele unterschiedliche Aspekte: Es ist ein Programmpaket zur Formelverarbeitung, es ist eine Programmiersprache, es ist ein Grafiksystem, kurz ein CAS, ein **C**omputer **A**lgebra **S**ystem. Es kann aber auch nur wie ein Taschenrechner bedient werden — auch so leicht. *Mathematica* ist recht umfangreich, doch das sollte Sie nicht schrecken. Ab Version 3 enthält *Mathematica* beachtenswerte Fähigkeiten zum Schreiben mathematischer Texte, ab Version 4 auch die Möglichkeit im Internet Seiten zu erstellen, ab Version 6 sind die Interaktionsfähigkeiten enorm erweitert.

Mathematica ist sehr gut beschrieben in dem offiziellen 1470-seitigen Handbuch:

> Wolfram, Stephen
>
> The *Mathematica* Book, 4th ed.
>
> Champaign: Wolfram Media, 1999
>
> Cambridge: Cambridge University Press, 1999

Dieses Buch wird oft einfach als Handbuch oder als *Mathematica*-Book zitiert.

Wenn Sie aber nur einen Internetzugang haben, dann gibt es The Mathematica Book online; das beschreibt allerdings Version 4 !

Sie erhalten beim Aufruf dieses Links daher auch folgenden Hinweis:

```
THIS IS DOCUMENTATION FOR AN OBSOLETE PRODUCT.
SEE THE DOCUMENTATION CENTER FOR THE LATEST INFORMATION.
```

Eine neue Version gibt es innerhalb einer installierten Version V7 von *Mathematica* als Virtual Book .

Über Help gelangen Sie in das Documentation Center von *Mathematica.* Der Umgang damit wird im Laufe des Kurses immer wieder geübt. Das Documentation Center ist auch im Internet zugänglich.

Ab 2007 gibt es das Wolfram Demonstrations Project. Auch das wird häufig benutzt werden.

Eine Liste von deutschspracigen Büchern, in denen auf *Mathematica* eingegangen wird, finden Sie hier. In unserer Bibliothek stehen die Bücher zu *Mathematica* unter 95 mat B 0.

Dieses Skript ist gespeichert unter

http://www.mathematik.uni-kassel.de/~rascha/Mathematica_Kurs/Skript_09_1.nb

Einige Beispiele zur Einstimmung und zum ersten Blick auf die Möglichkeiten von *Mathematica*.

Diese Beispiele können im Moment nicht immer vollständig erklärt werden. Wie der Editor zu bedienen ist, wird später erläutert. Kurz sei nur gesagt: "In[ ] := " und "Out[ ] = " mit ihrer jeweiligen Numerierung brauchen nicht eingetippt zu werden. Dann werden die SHIFT ↵ Tasten gedrückt und die Ergebniszeile erscheint. Manchmal lautet aber die Parole: Warte nur ein Weilchen. Auf geht's:

```
In[1]:= 5 + 7
Out[1]= 12
```

```
In[2]:= 1 / 3 + 2 / 5
```

Out[2]= $\frac{11}{15}$

In[3]:= `Simplify`$\left[a^2 + 2\,a\,b + b^2\right]$

Out[3]= $(a + b)^2$

Der *Mathematica* – Ausdruck `Simplify` erklärt sich selbst!

```
In[4]:= ((1 + I) / Sqrt[2]) ^100
Out[4]= -1
```

Eine andere Schreibweise:

In[5]:= $\left(\frac{1 + i}{\sqrt{2}}\right)^{100}$

Out[5]= -1

`I` steht für $\sqrt{-1}$. i ist eine weitere *Mathematica*-Form für die komplexe Zahl i . Statt $\sqrt{2}$ kann auch `Sqrt[2]` geschrieben werden.

```
In[6]:= N[Sqrt[10], 32]
Out[6]= 3.1622776601683793319988935444327
```

`N` steht für den numerischen Wert und der Parameterwert 32 gibt an, dass $\sqrt{10}$ mit 32 Dezimalstellen ausgegeben werden soll.

```
In[7]:= Timing[ N[ Zeta[1 / 2 + 14.13482 i], 20] ]
Out[7]= {0., -0.000011826 + 0.0000743032 i}
```

Die RIEMANNsche Zetafunktion $\sum_{n=1}^{\infty} \frac{1}{n^z}$ scheint kaum mehr Mühe als die Wurzelfunktion zu machen.

In[8]:= `Plot[ Abs[Zeta[1 / 2 + I t]], {t, 0, 15} ]`

Out[8]=



In[9]:= `FactorInteger[20 102 008]`

Out[9]= `{{2, 3}, {59, 1}, {42 589, 1}}`

Die Primfaktorzerlegung von 20 102 008 ist $2^3{\cdot}59^1{\cdot}42\,589^1$.

Bei Out[7], In[8] und Out[9] sind eben zum ersten Mal geschweifte Klammern aufgetaucht. *Mathematica* unterscheidet streng im Gebrauch von `[ ]`, `( )` und `{ }`. In eckigen Klammern `[ ]` werden die Argumente von Ausdrücken (engl.: *expressions*) eingeschlossen. In, Out, Simplify, Sqrt und Zeta (ζ-Funktion) bzw. \[Zeta] (Buchstabe ζ) sind in *Mathematica* Namen für vordefinierte Ausdrücke bzw. Kommandos.

In runden Klammern `( )` werden algebraische Terme eingeschlossen, wie etwa in Out[3] und in In[4]. Diese Unterscheidung von eckigen und runden Klammern bringt Vorteile in der Verdeutlichung von Ausdrücken, z.B.:

`c(1 + x)` entspricht $c{\cdot}(1+x)$,

`c[1 + x]` ist der Wert des Ausdruckes c bei $1+x$.

In geschweiften Klammern `{ }` werden Listen eingeschlossen, ähnlich wie beim Gebrauch von Mengenklammern.

`(*  *)` sind Kommentarklammern, die beliebig tief geschachtelt werden können.

In[10]:= `Table[ n!, {n, 1, 10} ]`

Out[10]= `{1, 2, 6, 24, 120, 720, 5040, 40 320, 362 880, 3 628 800}`

Die Tabelle der ersten 10 Fakultäten. So bekommen Sie das auf andere Weise:

In[11]:= `Array[ Factorial, 10 ]`

Out[11]= `{1, 2, 6, 24, 120, 720, 5040, 40 320, 362 880, 3 628 800}`

In[12]:= `m = Table[ 1 / (k + n + 1), {k, 1, 3}, {n, 1, 3} ];`

In[13]:= `m`

Out[13]= $\left\{\left\{\frac{1}{3}, \frac{1}{4}, \frac{1}{5}\right\}, \left\{\frac{1}{4}, \frac{1}{5}, \frac{1}{6}\right\}, \left\{\frac{1}{5}, \frac{1}{6}, \frac{1}{7}\right\}\right\}$

Dies ist eine mögliche Ausgabe der (3,3)-HILBERTmatrix *m*. Eine Matrix wird also als Liste von Listen - den Zeilen der Matrix - dargestellt. Auf dem Bildschirm ist aber auch die gewohnte Ausgabeform möglich:

In[14]:= `MatrixForm[ m ]`

Out[14]//MatrixForm=

$$\begin{pmatrix} \frac{1}{3} & \frac{1}{4} & \frac{1}{5} \\ \frac{1}{4} & \frac{1}{5} & \frac{1}{6} \\ \frac{1}{5} & \frac{1}{6} & \frac{1}{7} \end{pmatrix}$$

Eine einzelne Zeile der Matrix m :

In[15]:= `m[[1]]`

Out[15]= $\left\{\frac{1}{3}, \frac{1}{4}, \frac{1}{5}\right\}$

Das einzelne Element m_{13} :

In[16]:= `m[[1, 3]]`

Out[16]= $\frac{1}{5}$

Die Inverse von m wird berechnet und ausgegeben mit:

In[17]:= `MatrixForm[Inverse[ m ]]`

Out[17]//MatrixForm=

$$\begin{pmatrix} 300 & -900 & 630 \\ -900 & 2880 & -2100 \\ 630 & -2100 & 1575 \end{pmatrix}$$

Multiplikation der Inversen von m mit m . `%` steht für das zuletzt erhaltene Ergebnis:

In[18]:= `%.m`

Out[18]= `{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}`

In[19]:= `MatrixForm[%]`

Out[19]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Mathematica kann selbstverständlich auch symbolisch differenzieren:

In[20]:= $\mathtt{D[\, x^n \,,\ x\,]}$

Out[20]= $\mathtt{n\, x^{-1+n}}$

`D[`f`[`x`],` x `]` gibt den Wert der Ableitung von f nach x an:

In[21]:= $\mathtt{D\left[\, x^2\ Log[x + a],\ x\,\right]}$

Out[21]= $\frac{x^2}{a+x} + 2\, x\, \mathrm{Log}[a + x]$

Überlegen Sie sich, ob diese Formel allgemein gültig ist.

So sieht die Formel ein wenig gewohnter aus:

In[22]:= `TraditionalForm[%]`

Out[22]//TraditionalForm=

$$\frac{x^2}{a+x} + 2\log(a+x)\,x$$

Symbolische Integration verläuft äußerlich analog. Nun ja!

In[23]:= `Integrate[%, x]`

Out[23]= `x^2 Log[a + x]`

Bestimmte Integrale:

In[24]:= `Integrate[ Log[x] , {x, 1, a}, Assumptions → a > 1 ]`

Out[24]= `1 - a + a Log[a]`

In[25]:= $\int_1^a \mathtt{Log[x]}\, dx$

Out[25]=
```
If[Re[a] ≥ 0 || Im[a] ≠ 0, 1 + a (-1 + Log[a]),
  Integrate[Log[x], {x, 1, a}, Assumptions → ! (Re[a] ≥ 0 || Im[a] ≠ 0)]]
```

In[26]:= `TraditionalForm[%%]`

Out[26]//TraditionalForm=

$$\log(a)\,a - a + 1$$

Es gibt auch Integrale, die *Mathematica* nicht als Formel lösen kann:

In[27]:= `Integrate[ Sin[ Sin[x] ], {x, 0, 1} ]`

Out[27]= $\int_0^1 \mathtt{Sin[Sin[x]]}\, dx$

Nummerisch wird das Integral jedoch näherungsweise berechnet:

In[28]:= `N[ % ]`

Out[28]= `0.430606`

Beim Lösen von Gleichungen muss der Gebrauch der Zeichen = und == in *Mathematica* beachtet werden. Es ist wie in der Sprache C ; = wird als Zuweisungsoperator gebraucht (also wie := in Pascal) und == als Vergleichsoperator (also wie = in Pascal).

In[29]:= `Solve[ x^2 + 2 x - 7 == 0, x ]`

Out[29]= $\left\{\left\{x \to -1 - 2\sqrt{2}\right\}, \left\{x \to -1 + 2\sqrt{2}\right\}\right\}$

" `x →` " bedeutet: x wird der folgende Wert zugewiesen.

So erhalten Sie die nummerischen Werte:

In[30]:= `N[%]`

Out[30]= `{{x → -3.82843}, {x → 1.82843}}`

Mathematica löst Polynomgleichungen (mit einer Variablen) bis zum Grade vier und in speziellen Fällen noch einige mehr.

In[31]:= `Solve[ x^4 - 5 x^2 - 3 == 0, x ]`

Out[31]= $\left\{\left\{x \to -\sqrt{\frac{5}{2}+\frac{\sqrt{37}}{2}}\right\}, \left\{x \to \sqrt{\frac{5}{2}+\frac{\sqrt{37}}{2}}\right\}, \left\{x \to -i\sqrt{\frac{1}{2}\left(-5+\sqrt{37}\right)}\right\}, \left\{x \to i\sqrt{\frac{1}{2}\left(-5+\sqrt{37}\right)}\right\}\right\}$

In[32]:= `TraditionalForm[%]`

Out[32]//TraditionalForm=

$$\left\{\left\{x \to -\sqrt{\frac{5}{2}+\frac{\sqrt{37}}{2}}\right\}, \left\{x \to \sqrt{\frac{5}{2}+\frac{\sqrt{37}}{2}}\right\}, \left\{x \to -i\sqrt{\frac{1}{2}\left(-5+\sqrt{37}\right)}\right\}, \left\{x \to i\sqrt{\frac{1}{2}\left(-5+\sqrt{37}\right)}\right\}\right\}$$

Das sieht auch nicht viel anders aus!

Lineare Gleichungssyteme werden gelöst:

In[33]:= `Solve[{ x + y == 2, x + a y == 1 + a }, {x, y}]`

Out[33]= `{{x → 1, y → 1}}`

In[34]:= `Solve[ {x + y == 2, 2 x + 2 y == 4 }, {x, y} ]`

```
Solve::svars :
 Equations may not give solutions for all "solve" variables. >>
```

Out[34]= `{{x → 2 - y}}`

Inkonsistente Gleichungen:

In[35]:= `Solve[ {x == 1, x == 2} ]`

Out[35]= `{}`

Auch Differentialgleichungen können gelöst werden:

In[36]:= `DSolve[ y'[x] == a y[x], y[x], x ]`

Out[36]= $\{\{y[x] \to e^{a\,x}\,C[1]\}\}$

`C[1]` ist die Integrationskonstante.

Mathematica kennt die Potenzreihenentwicklungen vieler Funktionen:

In[37]:= `Series[` e^x`, {x, 0, 5} ]`

Out[37]= $1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + O[x]^6$

Jetzt sind die Namen einiger Funktionen in *Mathematica* vorgekommen: Sqrt, Log, Exp. Die Namen beginnen mit großen Buchstaben. *Mathematica* bleibt möglichst nahe an den allgemein üblichen Bezeichnungen von mathematischen Funktionen. Einige seien noch angeführt: Abs, ArcSin, Cos, Min, Max, aber auch AnimationDisplayTime oder IgnoreCase. Im Index of All Functions finden Sie alle diese Namen.

■ Grafik

Einige Beispiele sollen die Grafikmöglichkeiten demonstrieren.

In[38]:=
```
SetOptions[{Plot, ContourPlot},
  BaseStyle -> {FontFamily → "Times", FontSize → 12}];
```

Ein einfacher Graph einer Funktion:

In[39]:=
```
Plot[ Sin[x], {x,-π, π}]
```

Out[39]=



Eine erste Variation:

In[40]:=
```
Plot[Sin[ⅇ^x], {x, -π, π}, PlotLabel → Sin[ⅇ^x],
 LabelStyle → Directive[Plain, FontFamily → "Times"],
 FormatType → TraditionalForm]
```

Out[40]=



Eine Darstellung von $f(x, y) := x^3 - 3\,x y^2$:

In[41]:=
```
f[x_, y_] := x³ - 3 x y²
```

In[42]:= `Plot3D[f[x, y], {x, -1, 1}, {y, -1, 1}, BoxRatios → {1, 1, 1}]`

Out[42]=



Ein Höhenliniendiagramm dazu:

In[43]:= `ContourPlot[f[x, y], {x, -1, 1}, {y, -1, 1}, ColorFunction → Hue]`

Out[43]=



In[44]:= `ContourPlot[f[x, y], {x, -1, 1}, {y, -1, 1}, ColorFunction → Hue, PlotPoints → 101, Contours → 31]`

Out[44]=



Eine graphische Lösung der impliziten Gleichung $x^2 - y^4 = \sin(x^4\ y^4)$.

In[45]:= `ContourPlot[x^2 - y^4 == Sin[x^4 y^4], {x, 0, 2}, {y, 0, 1.5}, PlotPoints → 10]`

Out[45]=



In[46]:= `ContourPlot[x^2 - y^4 == Sin[x^4 y^4], {x, 0, 2}, {y, 0, 1.5}, PlotPoints → 50]`

Out[46]=



In[47]:= `ContourPlot[x^2 - y^4 == Sin[x^4 y^4], {x, 0, 4},`
`{y, 0, 2}, PlotPoints → 50, AspectRatio → Automatic]`

Out[47]=



Ein Vektorfeld, dazu muss das Paket "VectorFieldPlots`" geladen werden :

In[48]:= `Needs["VectorFieldPlots`"]`

In[49]:=
```
VectorFieldPlot[{Sin[x y], Cos[x y] - 1}, {x, 0, π},
  {y, 0, π}, PlotPoints → 16, ColorFunction → Hue,
  ScaleFunction → (#1/2 &), ScaleFactor → None, Frame → True,
  LabelStyle → Directive[Plain, FontFamily → "Times"]]
```

Out[49]=



Der Ausdruck Manipulate ist eine der wichtigsten Erweiterungen bei Version 6::

```
In[2]:= Manipulate[ Plot3D[ Sin[2 x] Sin[2 y] Cos[t],
    {x, 0, π}, {y, 0, π},
        PlotRange → {-1, 1},
        PlotPoints → 31,
        BoxRatios → {1, 1, 1},
        Ticks → None],
   {t, 0, π, π / 12}]
```

Out[2]=



Mittlerweile können Sie mit *Mathematica* mit einigen Betriebssystemen auch Töne erzeugen:

In[51]:= $\psi[\{\alpha\_, \beta\_\}, \mathcal{A}\_, f\_] := \mathrm{If}\left[\beta > \alpha, \mathrm{If}\left[\alpha < t < \beta, \mathrm{Sin}\left[\frac{\pi\,(t - \alpha)}{\beta - \alpha}\right] \mathcal{A}\, \mathrm{Sin}[f\, t], 0\right], 0\right]$

In[52]:=

$$\texttt{Play}\left[\texttt{Evaluate}\left[\texttt{N}\left[\texttt{Sin}\left[\frac{\pi\,t}{2}\right]\sum_{i=0}^{24}\psi\left[\left\{\frac{i}{12},\frac{i+4}{12}\right\},\sqrt{\left(\frac{i-24}{12}\right)^2},\,1000\,(i-12)\right]\right]\right],\ \{t,0,2\},\ \texttt{PlayRange}\to\texttt{All}\right]$$

Out[52]=